4.3 SIMULATION OF A COMPLETE CARDIOVASCULAR LOOP

A complete cardiovascular loop will now be set up in an uncontrolled form, that is, without the connections to the central nervous system (CNS), which provides much of the control of this system, and without consideration of the membrane connections to the body tissues that permit diffusion of plasma and of substances carried by the blood to and from such tissues. To do this we must devise a right heart model that will be much the same as the left heart model developed in Section 4.2, except that the peak systolic stiffness will need to be less by about a factor of four because of the smaller total capillary bed resistance in the lungs (about 170 CGS units) as compared to the systemic peripheral resistance (about 1300 CGS units). Such a model may be used to study blood volume shifts and other changes in response to parameter changes or defects (e.g., in heart valves).

Figure 4.3.1 shows a simple model of the uncontrolled cardiovascular system with right and left heart and associated arterial pathways much like the partial model of Fig. 4.2.5; the venous segments added to complete the loop are simple resistive segments. This model, which we will call pressure-flow zero (PF-0), is described by the equations in the ACSL program PF-0 below. This model has the following important characteristics:

1. Numerical values of parameters are entered in the equations except for the VSD conductance, GD = $1/R_{40}$, and the four ventricular stiffnesses, LS, LD, RS, and RD. Thus it is not possible to change anything except these five parameters at run time, unless they are introduced into the equations algebraically before compiling, with constants



Figure 4.3.1. A simple model of the complete cardiovascular loop shown in lumped circuit form (model PF-0).

shown separately. (Note that the three-letter designation scheme used in Section 4.2 is not used here, in order to be consistent with the first paper in which this model, with parameter values, was described (Rideout-72)).

2. Units are in the CGS system except that pressures are in medical units (mm Hg), requiring that a unit correction factor be included to multiply any pressure appearing in any equation. Thus for the flow f_3 (in ml/s) determined by the pressures across R_3

$$F3 = 1332.*(P2 - P3)/R3 = 8.9*(P2 - P3)$$

If we want to be able to change the pulmonary peripheral resistance R3 in this equation at run time it may be left in algebraic form and the command CONSTANT R3 = 150. added before compiling.

3. Numerical values of equation parameters correspond approximately to an adult male human weighing 70 kg at rest. Some allometric equations are known that enable some of the parameters given here to be roughly estimated for humans of other sizes or for dogs (see Section. 2.6).

4. The heart period TH is constant and corresponds to the radian frequency W of the sinusoid used in generating ventricular stiffness variations:

$$TH = 2.*\pi/W = 2.*3.14159/7.854 = 0.8 \text{ sec} \qquad (4.3.1)$$

5. This model may be used as given to study the effect of a VSD or of changes in the contractility of the ventricles (a myocardial infarction can be simulated by reducing SR and SL). As pointed out in item 2, other parameters can be changed at run time if included wherever they appear in algebraic form. In some cases (leaky or insufficient valves or atrial septal defects) new equations and/or changes in existing equations may be needed.

6. In this model no unstressed volumes need to appear, because no interchange between stressed and unstressed volume is assumed to occur. Thus stressed volumes are used throughout. Volume variables appear only in the ventricle, and here the unstressed volumes are assumed to be zero. (A more general approach to blood volume is used in the next model, PF-1.)

```
PROGRAM PF-0
 DYNAMIC
   CONSTANT TF = 4.0
   TERMT(T .GE. TF)

  DERIVATIVE
     Algorithm IALG = 4  $' 2nd order RK'
     Maxterval MAXT = .004
     Cinterval CINT = .02
     Nsteps NSTP=1
   Constant RD=.044, RS=.30, W=7.854

     F0=133.2*BOUND(0.,2000.,P10-P0)      $'Right Heart'
     Q0=INTEG(F0-F1+F40,154.)
     S0=RS*BOUND(0.,2.,SIN(W*T))+RD
     P0=S0*Q0

     P1=.75*INTEG(F1-F2,23.)              $'Pul. Art. 1'
     F1=133.2*BOUND(0.,2000.,P0-P1)

     F2=1000.*INTEG(P1-P2-F2/50.,.043)    $'Pul. Art. 2'
     P2=.25*INTEG(F2-F3,52.)

     P3=.15*INTEG(F3-F4,41.)              $'Lung Capillaries'
     F3=8.9*(P2-P3)
```



```
   Constant LD=.033, LS=1.5               $'Left Heart'
     Q4=INTEG(F4-F5-F40,158.)
     S4=LS*BOUND(0.,2.,SIN(W*T))+LD
     P4=S4*Q4
     F4=133.2*BOUND(0.,2000.,P3-P4)

   Constant GD=0.0                        $'VSD'
     F40=1332.*GD*(P4-P0)

     P5=1.5*INTEG(F5-F6,64.)              $'Ascend. Aorta'
     F5=100.*BOUND(0.,2000.,P4-P5)

     F6=1000.*INTEG(P5-P6-.005*F6.,016)   $'Descend. Aorta'
     P6=.75*INTEG(F6-F11-F7,129.)

     F7=1000.*INTEG(P6-P7-.02*F7,.0024)   $'Abdom. Aorta'
     P7=.562*INTEG(F7-F8-F13,172.)

     P8=.0903*INTEG(F8-F9,97.)            $'Leg Arteries'
     F8=.25*(P7-P8)

     P9=.075*INTEG(F9-F10+F12+F13,113.)   $'Veins'
     F9=100.*(P8-P9)
     P10=.15*INTEG(F10-F0,53.)
     F10=100.*(P9-P10)
     P11=.375*INTEG(F11-F12,145.)

     F11=P6-P11                           $'Upper Body'
     F12=P11-P9
     F13=.25*(P7-P9)                      $'Internal Organs'
     CO=  REALPL(TCO,F12+F13+F8,COIC)     $'Ave. Cardiac Output'
     FP=  REALPL(TFP,F2,FPIC)             $'Ave. Pul. Flow'
     FVSD=REALPL(TFVSD,F40,FVSDIC)        $'Ave. VSD Flow'
   Constant TCO=4.,TFP=4.,TFVSD=4., . . .
    COIC=70.,FPIC=70.,FVSDIC=50.
  END $ 'of Derivative'

 End $ 'of Dynamic'

End $ 'of Program'
```

If this model is compiled and run, the first variables of interest are the ventricular volumes, Q0 and Q4, shown in Fig. 4.3.2a. Note that the volume pumped out of each ventricle per beat is approximately 90 ml, which means that the cardiac output is 90/0.8 = 112 ml/s. Peak volumes are about the same in each ventricle, approximately 160 ml, and thus the ejection fraction for each ventricle is 90/160 = 0.56.

The model goes into a repetitive steady state at about the second beat. This occurs because the initial conditions are nearly right to start the model at the end of diastole, in steady state. These initial conditions were determined by first guessing at their values and then running the model until it reached steady state and all starting transients had disappeared. The model was stopped at the end of diastole, and the outputs of all variables obtained from integrators were obtained using the DISPLY command. These values were then used as integrator initial conditions.

Figures 4.3.3a shows the right ventricular pressure P0, together with the pulmonary artery pressure, P1; the corresponding pressures for the left ventricle are shown in Fig. 4.3.3c. Figures 4.3.3b and 4.3.3d show the important pressure-volume plots for the right and left ventricles.

Note that in Fig. 4.3.3c the amplitude of the left ventricular pressure is close to an expected peak value of 120 mm Hg, whereas the right ventricle peak in Fig. 4.3.3a is somewhat higher than the normal 18 mm Hg. This might be somewhat corrected by lowering the systolic stiffness, SR. Also, the aortic pressure has a high diastolic value; this might be



Figure 4.3.2. (a) Ventricular volumes during normal steady-state operation of model PF-0.
(b) Inflow, F_4, and outflow, F_5, for the left ventricle.



Figure 4.3.3. (a) Right ventricular pressure P0 and pulmonary artery pressure P1.
(b) Pressure versus volume locus for the right ventricle.
(c) Left ventricular and aortic pressures, P4 and P5.
(d) P-V locus for the left ventricle.

corrected by reducing the arterial compliances, C5, C6, and C7, so that the smaller resultant time constants give a faster decay of P5 after the aortic valve closes. Finally, the cardiac output is somewhat high, probably as a result of excess volume; choice of initial volumes will be considered later in model PF-1.

The ventricle pressure-flow loci (see Fig. 4.3.3b) are useful to the modeler in several ways; the stroke volume is easily obtained, as shown, as well as the peak pressure and ejection fraction. The ventricular stiffness slopes have been drawn in for the case of the left ventricle.

If a ventricular septal defect (VSD) is introduced into model PF-0 by setting the ventricle-to-ventricle conductance GD = 0.004, a number of waveforms change, as shown in Fig. 4.3.4, and overall performance is reduced as evidenced by a smaller cardiac output. The flows F5 (aorta) and F40 (VSD) may be seen to be comparable in peak amplitude and more nearly equal in the area under the flow pulses in the two cases. Thus it appears that about half of the left ventricular output is returned to the right ventricle through the defect in this case. (Note that there is a small settling transient after startup in the variables in the VSD case, because the initial conditions are still set for the normal case.)



Figure 4.3.4. Waveforms in model PF-0 with VSD.
(a) Top: f_5, aortic flow, and below: f_{40}, VSD flow.
(b) Top: right ventricular average volume Q_0 is increased and volume variation reduced by the effect of the VSD. Below: left ventricular average volume Q_4 is decreased and volume variation increased by the VSD.

When a VSD is present, the cardiac output cannot be determined from the volume variation in either ventricle, because we no longer have a simple blood flow loop. However, it is possible to find the average cardiac output (CO) by filtering F5, or better, by filtering the sum of less pulsatile flows, using the command

```
CO = REALPL(TCO, F8 + F11 + F13, COIC)
```

where TCO = 5.0, (a 5-sec time constant) and COIC = 70.0, (a first guess at cardiac output, in ml/s). This will show that cardiac output is 78 ml/s. It is also possible to filter F40 and show that the average VSD flow is 76 ml/s. If the filtered average pulmonary capillary flow F3 is determined, it may be shown to be 154 ml/s, which is the sum of the other two averaged flows, as would be expected.